

```

{ Problem ustawiania hetmanów na szachownicy }

const
  size = 8;      { rozmiar szachownicy }
  size2 = 15;    { liczba przekątnych size2 = 2*size-1 }
var
  i : integer;
  t : array [1..size] of integer;          { tablica zawierająca rozwiązanie }
  prawa,lewa : array [1..size2] of boolean; { przekątne prawe i lewe }

procedure wypisz; { wypisz rozwiązanie zawarte w tablicy t }
var
  i,j : integer;
begin
  for j:=1 to size do begin
    for i:=1 to t[j]-1 do write(' ');
    write(' $');
    for i:=t[j]+1 to size do write(' ');
    writeln;
  end;
  writeln;
  readln;      { zatrzymaj się po wypisaniu rozwiązania, ENTER - dalej }
end;

procedure ustaw( kol : integer ); { ustaw hetmana w kolumnie kol }
var
  wrs : integer;

begin
  if kol>size then wypisz { jeżeli ustawione wszystkie wypisz rozwiązanie }
  else
    for wrs:=1 to size do { próbuj ustawić w wierszu wrs }
      if (t[wrs]=0) and { czy wolny wiersz }
        not prawa[wrs+kol-1] and { czy wolna prawa przekątna }
        not lewa[size-kol+wrs] { czy wolna lewa przekątna }
      then begin
        { ustaw hetmana w kolumnie kol i wierszu wrs }
        t[wrs]:=kol; prawa[wrs+kol-1]:=true; lewa[size-kol+wrs]:=true;
        { przejdź do ustawiania hetmana w następnej kolumnie }
        ustaw(kol+1);
        { wycofaj ustawienie hetmana w kolumnie kol i wierszu wrs }
        t[wrs]:=0; prawa[wrs+kol-1]:=false; lewa[size-kol+wrs]:=false;
      end
    end;

begin
  for i:=1 to size do t[i]:=0;
  for i:=1 to size2 do begin prawa[i]:=false; lewa[i]:=false end;
  ustaw(1); { ustaw pierwszego hetmana }
end.

```